

U. S. DEPT. OF INTERIOR  
GEOL. SURVEY

Programs for filing and plotting U-Pb isotope data for concordia diagrams,  
using an HP-9830 computer and HP-9862 plotter

by Kenneth R. Ludwig

Open-File Report 82-386  
1982

This report is preliminary and has not been reviewed for conformity with U.S.  
Geological Survey editorial standards. Any use of trade names is for  
descriptive purposes only and does not imply endorsement by the USGS.

## INTRODUCTION

### Hardware Required

These programs, modified from earlier programs written for the Hewlett-Packard (HP) model 9831 computer and 9872 plotter combination (Ludwig, 1979), are specifically written for an HP-9830 computer with an HP-9862 plotter and are optimized for these slower and less powerful machines. The programs require a Hewlett-Packard model HP-9830 desktop computer with 8K words of memory, a plotter ROM and string-variable ROM, an HP-9862 plotter, and a printer set to a select code of 15. Three files must be created in the tape cassette to run the program. The first two should be 4100 words in length (for the plotter segments, parts 1 and 2), and the third should be 1100 words in length (the data-file program). Additional files of 500 words are used to store the data. Note that the file numbers of the two plotter segments are given in line number 90 of the first plotter-program segment. These must be changed to the actual file numbers used before the program is stored.

### Purpose and Scope of Programs

These programs are designed to store and plot U-Pb isotope data in the form  $^{206}\text{Pb}(\text{radiogenic})/^{238}\text{U}$ ,  $^{207}\text{Pb}(\text{radiogenic})/^{235}\text{U}$ , the errors and error-correlations associated with these two parameters, and alphanumeric sample-names for each data point. Once stored in a data file, the data can be edited or added to and re-stored on tape. The plotting segment of the program will draw a concordia diagram (of the conventional form, i.e.,  $^{206}\text{Pb}/^{238}\text{U}$  versus  $^{207}\text{Pb}/^{235}\text{U}$ ) with user-specified limits, and plot data either from data files or keyboard entry using a variety of symbols or an error-ellipse. These data can then be used to define a least-squares regression line, with intercepts and intercept uncertainties of the line with the concordia curve calculated by the program.

## OPERATION OF DATA FILE PROGRAM

Load the appropriate file into the computer memory. The first display (after executing a RUN command) is CREATE(1) ADD(2) EDIT(3) PRINT(4)?. (1) permits creation of a new data file (no previous data); (2) permits adding to an existing file; (3) permits editing of existing data, and (4) permits printing out of all of the data in a data file.

### CREATE option:

The first display, FILE NAME?, requests a name (no more than 20 characters long) for the new data file. The next display, SAMPLE NAME?, request a name (no more than 12 characters long) for the data set. The next display, 6/8, ERR, 7/5, ERR, CORREL? requests five numeric values for this data set:  $^{206}\text{Pb}_{\text{rad.}}/^{238}\text{U}$ ., 206/238 error,  $^{207}\text{Pb}_{\text{rad.}}/^{235}\text{U}$ , 207/235 error, and the error-correlation between 206/238 and 207/235. The errors must be in percent and at the 2 $\sigma$  level. The error correlation must be between -1 and +1, and should be consistent with the relation

$$\rho(x,y) = \frac{\Sigma_x^2 + \Sigma_y^2 - \Sigma_z^2}{2\Sigma_x\Sigma_y}$$

where  $x = ^{207}\text{Pb}/^{235}\text{U}$ ,  $y = ^{206}\text{Pb}/^{238}\text{U}$ ,

$z = ^{207}\text{Pb}/^{206}\text{Pb}$ , and  $\rho(x,y)$  is the error-correlation between  $x$  and  $y$  (Ludwig, 1980).

The computer will then request the name and data for the next data set, and so on, until the user escapes the cycle by answering the SAMPLE NAME? query with a space (space bar). The display will then be STORE IN FILE#?. If you do not wish to store the data on tape yet, enter a space. Otherwise,

enter the file number (must be at least 400 words marked length) of the file to receive the data.

#### ADD OPTION:

The first display will be DATA FROM FILE#?. If the data is already in memory, enter a space. Otherwise, enter the file number of the data-file of interest. The additional data will then be requested as above.

#### EDIT OPTION:

Enter the set number to be edited when requested, and enter the edited data as a new entry. When all required sets have been edited, answer the EDIT SET#? query with a space.

#### PRINT OPTION:

Operation is straightforward. Remember, it is not necessary to specify a file number for loading of data if the data is already in memory, nor for storing of data if the data in memory is not ready for permanent storage.

#### OPERATION OF PLOTTING PROGRAM

Before running this program, place a blank piece of paper on the plotter, and position the lower-left and upper-right plotting limits with the appropriate buttons and knobs on the plotter. Load the file containing the first segment of the program.

The first display after executing a RUN command is DATA-FILE#? If you wish to plot at least some data from a data file, enter its file number (making sure the tape cassette with that file is in the computer). If the data-file is already in memory, you can avoid an unnecessary re-load of data by entering 100. If all data are to be keyboard-entered, answer the query with a space. The next display, HEIGHT, WIDTH OF PLOT (CM)? requests the vertical and horizontal dimensions of the plot, in centimeters, as defined by the "lower left" and "upper right" controls on the plotter itself. The next

display, TITLE?, requests a title (no more than 38 characters long) for the plot, which will be lettered in the upper left-hand corner of the plot. The next display, AGE LIMITS, M.Y. (MIN., MAX.)?, requests the maximum and minimum ages to be defined by the concordia curve (e.g. 0, 2350). If you would rather define the plot limits in terms of X- and Y-limits, answer this query with 0, 0 and the display will be 7/5 AND 6/8 LIMITS?. Enter these limits in the form  $^{207}\text{Pb}/^{235}\text{U}$  lower limit,  $^{207}\text{Pb}/^{235}\text{U}$  upper limit,  $^{206}\text{Pb}/^{238}\text{U}$  lower limit,  $^{206}\text{Pb}/^{238}\text{U}$  upper limit (4 values).

The plotter will then draw the plot box, selecting appropriate tick intervals, label the ticks (centered and left-justified) and axes, and draw the concordia curve with labeled ticks.

The next display will be DATA SYMBOL OR CODE?, followed in 1.5 seconds by P=POLYGON, E=ERROR ELLIPSE?. An input of P permits plotting of data using any regular polygon for a symbol, as defined by the # SIDES, ROTATION, SIZE (1-10)? query. Enter 3 values in response, the first indicating how many sides to the polygon (e.g. 3 yields a triangle), the second indicating the rotational orientation of one of its vertices (measured in degrees anti-clockwise from the 3 o'clock orientation), the third indicating relative size (sizes of 2 to 4 are most useful). A lower-case P will give an open polygon whereas an upper-case P will give a solid polygon.

An input of E will give an error ellipse as the plotting symbol, with the dimensions and shape defined (at the  $2\sigma$  level) by the errors and error correlations of the data-points. Again, lower-case E yields an open symbol, whereas upper-case E gives a solid symbol.

An input of L will permit lettering of any phrase on the plot, and is followed by a LETTER SIZE (.5-2), ANGLE? display. Input two values, the first being the relative size of the letters, the second the anti-clockwise angle

from the horizontal at which the phrase is to be lettered. The next display will be a question mark, at which point the four arrow-symbol keys at the top of the keyboard are used to position the pen for lettering. When the pen is properly positioned, press the STOP key. In answer to the PHRASE? display, type in the phrase to be lettered, then press the EXECUTE key.

Any other symmetrical symbol, such as X, +, \*, or 0 may also be used as a plotting symbol.

After choosing the plotting symbol, the display will be DATA FILE SETS? if a data-file was loaded into memory. If one number is entered, the data for that set number of the data-file in memory will be plotted. If two numbers separated by a comma are entered, all data sets between the two numbers (inclusive) will be plotted (e.g., 5, 8 will result in sets 5, 6, 7 and 8 being plotted). To enter data from a data file not in memory, preface the set numbers by the letters DF followed by the data-file number. Thus the input DF7,2,5 would give sets 2 through 5 from data file 7. To enter data from the keyboard, answer the DATA-FILE SETS? query with a space.

If no data-file is loaded into memory, or if data are to be input from the keyboard instead of from a data-file, the display will be 6/8, ERR, 7/5, ERR, CORREL.?. Enter five values, as defined in the data file segment instructions, or enter a space if no more data is to be entered. Remember, errors must be entered in percent and at the  $2\sigma$  level.

The display SET TO BE CORRECTED (0=OK)? requests the set number (as defined in the just-printed list) of any incorrectly entered set. If the response is a nonzero number, data for that set (from the keyboard only) will be re-requested. If the response is 0, the data will be plotted.

The next display will be YORKFIT? 1=YES?. Enter a zero if you do not wish a regression line to be calculated for the plotted data, or if the data

are to be combined with additional data for regression. If the Yorkfit (regression calculation) is declined, the display will be DATA TO BE RETAINED? 1=YES? Enter 1 if the plotted data are to be combined with more data (perhaps with a different plotting symbol or color of pen) for an eventual Yorkfit.

If a Yorkfit is requested, the next segment of the program will be loaded into memory, and an initial York regression calculated in the standard way (York, 1969), with points weighted according to analytical error and slope, and intercept errors of the Yorkfit line propagated from the analytical errors (MODEL 1). If a chi-square calculation indicates that the assigned errors will result in at least the observed degree of scatter more than 30 percent of the time, the MODEL 1 fit will be automatically accepted. If the calculated probability is less than 30 percent, the regression will be recalculated (MODEL 2) assuming equal weighting of the points and zero correlation of errors (in other words, assuming that nothing is known about the reason for the excess scatter). If the chi-square probability is between 30 percent and 5 percent, the user may choose which model calculation to accept. If the probability is less than 5 percent only the MODEL 2 calculation will be used.

The plotter will then draw the Yorkfit line and solve for the intercepts of the line with the concordia curve, and the intercepts of the 95 percent confidence error-envelope about the line (Ludwig, 1980). Any insoluble intercepts will be printed out as zero age. The next query will be ENTER 1 FOR ERROR-ENVELOPE PLOT. If 1 is entered, the 95 percent confidence error-envelope about the least-squares line will be drawn (useful for identifying outlier points and for understanding the age uncertainties).

## REFERENCES

- Ludwig, K. R., 1979, A computer program in Hewlett-Packard BASIC for plotting and processing U-Pb isotope date in "concordia" diagrams: U.S. Geol. Survey Open-File Report 79-1692, 32 p.
- Ludwig, K. R., 1980, Calculation of uncertainties of U-Pb isotope data: Earth Planet. Science Letters, v. 46, p. 212-220.
- York, Derek, 1969, Least squares fitting of a line with correlated errors: Earth Planet. Science Letters, v. 5, p. 320-324.



# APPENDIX I--List of

## DATA-FILE PROGRAM

```

10 COM DS[20,5],F$[20],NI[20,6]
20 REM DATA-FILE PROGRAM FOR HP-9830/9862 CONCORDIA PLOTTER- TAKES 20 SETS
30 REM K.R. LUDWIG, 1982
40 DIM N$[12],L$[58],A$[10],Z$[10]
50 L$=" ABCDEFGHIJKLMNOPQRSTUVWXYZ.0123456789!#$%&'();:,<>?+-*/^="
60 Z$="0123456789"
70 DISP "CREATE(1)ADD(2)EDIT(3)PRINT(4)";
80 INPUT F
90 IF F>1 THEN 440
100 DISP "FILE NAME";
110 INPUT F$[1,20]
120 FOR I=1 TO 20
130 FOR J=1 TO 5
140 D[I,J]=0
150 N[I,J]=0
160 NEXT J
170 NEXT I
180 N[20,5]=0
190 FOR I=1 TO 20
200 GOSUB 230
210 NEXT I
220 GOTO 330
230 DISP "SAMPLE#"I"NAME";
240 INPUT N$[1,12]
250 IF N$[1,12]=" " THEN 330
260 GOSUB 330
270 DISP TAB1+FNAI"6/8, ERR, 7/5, ERR, CORREL.";
280 INPUT D[I,1],D[I,3],D[I,2],D[I,4],D[I,5]
290 IF ABSD[I,5] <= 1 THEN 320
300 PRINT TAB90"** CORRELATION MUST BE BETWEEN -1 AND +1 ***"LIN2
310 GOTO 270
320 RETURN
330 DISP "STORE IN FILE#";
340 INPUT A$
350 IF FNC0<0 THEN 70
360 STORE DATA G
370 GOTO 70
380 DISP "EDIT SET#";
390 INPUT A$
400 I=FNC0
410 IF I<0 OR I>20 THEN 330
420 GOSUB 230
430 GOTO 380
440 DISP "DATA FROM FILE#";
450 INPUT A$
460 IF FNC0<0 THEN 490
470 LOAD DATA G
480 PRINT "FILE NAME: "F$,LIN2
490 GOTO F-1 OF 880,380,500
500 PRINT "SET#"TAB22"206/238 - %ERR 207/235 - %EPR CORREL."LIN2

```

```

510 FOR I=1 TO 20
520 IF D[I,1]=0 THEN 560
530 Y=FNBI
540 PRINT I;N$;TAB20;D[I,1];TAB32;D[I,3];TAB42;D[I,2];TAB53;D[I,4];TAB63;D[I,5]
550 NEXT I
560 PRINT TAB90,TAB90
570 GOTO 70
580 DEF FNC(A)
590 IF POS(Z$,A$(1,1)) THEN 620
600 G=-1
610 GOTO 630
620 G=VAL(A$)
630 RETURN G
640 DEF FNA(K)
650 FOR J=1 TO INT(0.5+LEN(N$)/2)
660 N[K,J]=100*POS(L$,N$(2*J-1,2*J-1))+POS(L$,N$(2*J,2*J))
670 NEXT J
680 FOR J=J TO 6
690 N[K,J]=0
700 NEXT J
710 RETURN 0
720 DEF FNB(K)
730 N$=""
740 FOR J=1 TO 6
750 P=INT(N[K,J]/100)
760 IF P=0 THEN 820
770 Q=N[K,J]-100*P
780 N$(2*J-1,2*J-1)=L$(P,P)
790 IF Q=0 THEN 820
800 N$(2*J,2*J)=L$(Q,Q)
810 NEXT J
820 RETURN K
830 FOR J=12 TO 1 STEP -1
840 IF NOT POS(N$(J,J)," ") THEN 870
850 N$=N$(1,J-1)
860 NEXT J
870 RETURN
880 FOR K=1 TO 20
890 IF D[K,1]=0 THEN 910
900 NEXT K
910 IF K#21 THEN 940
920 PRINT LIN1"*** THIS DATA-FILE IS FULL ***"LIN2
930 GOTO 70
940 FOR I=K TO 20
950 GOSUB 230
960 NEXT I
970 GOTO 330
980 END

```

# APPENDIX II--LIST OF PLOTTER PROGRAM

## First Segment:

```

10 COM DS[20,5],F$[20],NI[20,6]
20 REM CONCORDIA PLOTTER FOR 8K HP-9830/9862, PART 1, K.R. LUDWIG, 1982
30 REM CAN TAKE DATA FROM 20-SET DATA-FILES
40 GOTO 110
50 GOTO 2240
60 DIM AS[20],GS[20],QS[20],RS[20],ZS[20]
70 DIM A$[38],C$[12],XS[20],YS[20],L$[58],N$[12],FI[2]
80 REM FILE #S OF PARTS 1 AND 2
90 DATA 0,1
100 DATA 0,0,0,9.8485E-04,1.55125E-04,"207206235238"
110 READ F[1],F[2],Q1,P1,D,L5,L8,C$
120 L$=" ABCDEFGHIJKLMNOPQRSTUVWXYZ.0123456789!#$%&'();:,<>?+-*/^="
130 DEG
140 GOSUB 160
150 GOTO 1110
160 DISP "DATA-FILE #";
170 INPUT A$
180 IF FNZ1<0 OR F=100 THEN 210
190 LOAD DATA F
200 PRINT LIN1"DATA FROM FILE"F;TAB20;F$;LIN2
210 F0=F
220 RETURN
230 FIXED 0
240 RETURN
250 FIXED 1
260 RETURN
270 FIXED 2
280 RETURN
290 FIXED 3
300 RETURN
310 FIXED 4
320 RETURN
330 DEF FNA(T)=EXP(L5*T)-1
340 DEF FNB(T)=EXP(L8*T)-1
350 DEF FNC(T)=L8*EXP((L8-L5)*T)/L5
360 DEF FND(X)=LOG(1+X)/L5
370 DEF FNE(Y)=LOG(1+Y)/L8
380 DEF FNF(X)
390 S=0
400 N=X
410 N=10*(N-INTN)
420 IF N=0 THEN 450
430 S=S+1
440 GOTO 410
450 RETURN S
460 DEF FNK(X)=1+(X#INTX)+(X >= 1)*FNG(X+ NOT X)+FNFX

```

```

470 DEF FNG(X)=INTLGTX
480 DEF FNH(X)
490 Q=10^FNGX/8
500 IF X/Q <= 12 THEN 530
510 Q=2*Q
520 GOTO 500
530 Q2=Q/10^FNGQ
540 IF Q2=INTQ2 THEN 560
550 Q=INTQ2*10^FNGQ
560 RETURN Q
570 DEF FNJ(X)=- ( NOT X)-(X#0)*(FNKX+J-FNFX+(X=INTX)*(J#0))
580 DEF FNO(F)=Y[I]+(-B3*X+F*SQRFl)/(2*S3)
590 DEF FNN(X)
600 IF X=0 THEN 630
610 LABEL (*,F4,1.7,0,P)
620 CPLOT -P1,0
630 LABEL (*,F3,1.7,90*X,P)
640 CPLOT -5,0.9*(1-3* NOT X)
650 LABEL (*)C$[1+3*X,3+3*X];
660 CPLOT 0,-0.3
670 LABEL (*)"P /";
680 CPLOT 0,0.3
690 LABEL (*)C$[7+3*X,9+3*X];
700 CPLOT 0,-0.3
710 LABEL (*)"U";
720 CPLOT -6.3,0
730 LABEL (*)"I";
740 CPLOT -0.7,0
750 LABEL (*,0.6*F3,1.7,90*X,P)"O";
760 PLOT A1,A2+S6/2,1
770 RETURN 0
780 DEF FNQ(X)
790 S=150+50*((I1*I3)>1000)-70*((I1*I3)<400)
800 FOR I=1 TO Z6/A8
810 IPLOT X*A8,0
820 IPLOT 0,X*S6/S
830 IPLOT 0,-X*S6/S
840 NEXT I
850 RETURN 0
860 DEF FNM(X)
870 FOR I=1 TO S6/A5
880 IPLOT 0,X*A5
890 IPLOT -X*Z6*P/S,0
900 IPLOT X*Z6*P/S,0
910 NEXT I
920 RETURN 0
930 DEF FNR(I)
940 N$=""
950 FOR D=1 TC 6
960 B=INT(N[I,D]/100)

```

```

970 IF B <= 0 OR B>58 THEN 1030
980 F=N[I,D]-100*B
990 N$[2*D-1,2*D-1]=L$[B,B]
1000 IF F=0 THEN 1030
1010 N$[2*D,2*D]=L$[F,F]
1020 NEXT D
1030 RETURN 0
1040 DEF FNZ(I)
1050 IF POS(L$[28+I,38],A$[1,1]) THEN 1080
1060 F=0-I
1070 GOTO 1090
1080 F=VAL(A$)
1090 A$=A$[1+POS(A$," ")]
1100 RETURN F
1110 DISP "HEIGHT, WIDTH OF PLOT (CM)";
1120 INPUT I1,I3
1130 P=I1/I3
1140 F3=3.5*(1+(I1<13)/3-(I1>23)/3)
1150 F4=2*F3/3
1160 DISP "TITLE";
1170 INPUT A$
1180 DISP "AGE LIMITS, M.Y. (MIN., MAX.)";
1190 INPUT H1,B3
1200 IF B3>H1 THEN 1280
1210 DISP "7/5 AND 6/8 LIMITS";
1220 INPUT A1,B1,A2,B2
1230 B3=FND(B1-(B1-A1)/100)
1240 IF FNBB3<B2 THEN 1260
1250 B3=FNEB2
1260 H1=FNDA1
1270 GOTO 1390
1280 B1=FNAB3
1290 B2=FNBB3
1300 A8=FNH(B1-FNAH1)
1310 A5=FNH(B2-FNBH1)
1320 FOR A1=0 TO B1 STEP A8
1330 IF FND(A1+A8)>H1 THEN 1350
1340 NEXT A1
1350 B1=B1+A8
1360 FOR A2=0 TO B2 STEP A5
1370 IF FNE(A2+2*A5)>H1 THEN 1410
1380 NEXT A2
1390 A8=FNH(B1-A1)
1400 A5=FNH(B2-A2)
1410 A9=FNH((B3-H1)/(1+((B3-H1)>3000)))
1420 FOR Y=A2 TO B2 STEP 2*A5
1430 IF FNKY<P1 THEN 1450
1440 P1=FNKY
1450 NEXT Y
1460 B1=A1+A8*INT(1+0.99*(B1-A1)/A8)

```

```

1470 B2=A2+A5*INT(1+0.99*(B2-A2)/A5)
1480 Z6=B1-A1
1490 S6=B2-A2
1500 P4=0.037*F3*I1+(P1+0.5)*F4*I1/170
1510 Z1=3.5*F4*I1/100
1520 P6=5*Z1/7
1530 C=F4*I1/100+0.038*F3*I1
1540 Q2=I3-P4-Z1
1550 P2=A1-Z6*P4/Q2
1560 L=B1+Z6*Z1/Q2
1570 B5=I1-P6-C
1580 E=A2-S6*C/B5
1590 P0=B2+S6*P6/B5
1600 P5=L-P2
1610 P6=P0-E
1620 SCALE P2,L,E,P0
1630 PLOT A1,A2
1640 J=FNQ1+FNMI+FNQ(-1)+FNM(-1)
1650 PEN
1660 FOR X=A1 TO B1 STEP 2*(1+(I3<18))*A8
1670 IF FNFX<J THEN 1690
1680 J=FNFX
1690 NEXT X
1700 FOR X=A1 TO B1 STEP 2*(1+(I3<18))*A8
1710 LABEL (*,F4,1.7,0,P)
1720 PLOT X,A2,1
1730 CPLOT FNJX/2-0.7,-1
1740 GOSUB 1+J*(X#0) OF 230,250,270,290,310
1750 LABEL (*)X;
1760 NEXT X
1770 J=0
1780 FOR Y=A2 TO B2 STEP 2*A5
1790 IF FNFY<J THEN 1810
1800 J=FNFY
1810 NEXT Y
1820 FOR Y=A2 TO B2 STEP 2*A5
1830 PLOT A1,Y,1
1840 CPLOT FNJY-1.4,-0.3+0.1*(Y=A2)
1850 GOSUB 1+J*(Y#0) OF 230,250,270,290,310
1860 LABEL (*)Y;
1870 NEXT Y
1880 PLOT A1+Z6/2,A2,1
1890 FIXED 0
1900 F5=F4*3/4+FNN0+FNN1
1910 K=8*A9
1920 C=0
1930 FOR H1=C TO B3 STEP K
1940 IF FNAH1>A1 AND FNBH1>A2 THEN 1960
1950 NEXT H1
1960 IF K=A9 THEN 2000

```

```

1970 C=H1-K
1980 K=A9
1990 GOTO 1930
2000 FOR T=H1-A9 TO B3 STEP A9/8
2010 X=FNAT
2020 Y=FNBT
2030 IF X<A1 OR Y<A2 OR Y>B1 OR Y>B2 THEN 2180
2040 PLOT X,Y
2050 GOTO NOT T+2*(T/A9#INT(T/A9)) OF 2170,2180
2060 Q=ATN(FNCT*P*P5/P6)-90
2070 LABEL (*,F5,1.7,Q,P)
2080 CPLOT -0.3,-0.3
2090 LABEL (*)"-";
2100 IF D THEN 2160
2110 X1=2+FNGT
2120 IF 0>(X-A1)*Q2/Z6-COSQ*F5*I1*(X1+4)/170 THEN 2160
2130 IF 0>(B2+A2-Y)*B5/S6+SINQ*F5*I1*(X1+4)/100 THEN 2160
2140 CPLOT -X1-1.3,0
2150 LABEL (*)T;
2160 PLOT X,Y
2170 D= NOT D
2180 NEXT T
2190 PLOT A1,B2,1
2200 LABEL (*,F4,1.7,0,P)
2210 CPLOT 1.6,-1.6
2220 LABEL (*)A$;
2230 PRINT "INPUT L TO LETTER"TAB44"CONT 0 TO RESTART DATA-PLOT"LIN1
2240 PLOT L,P0,1
2250 K=I=1
2260 N=U=V=0
2270 STANDARD
2280 C$="lLpPeE+*"
2290 DISP "DATA SYMBOL OR CODE"
2300 WAIT 1500
2310 DISP "P=POLYGON, E=ERROR-ELLIPSE";
2320 INPUT B$[1,1]
2330 A=POS(C$,B$)
2340 C=0
2350 GOTO A OF 3500,3500,3280,3280
2360 IF F0<0 THEN 2450
2370 DISP "DATA-FILE SETS";
2380 INPUT A$
2390 IF A$[1,2]#"DF" THEN 2420
2400 A$=A$[3]
2410 GOSUB 180
2420 U=FNZO
2430 V=(U#0)*FNZO
2440 IF V AND U THEN 2590
2450 C=1
2460 GOTO 2640

```

```

2470 GOSUB 2490
2480 GOTO 2590
2490 DISP "6/8, ERR, 7/5, ERR, CORREL.";
2500 INPUT A$
2510 IF A$=" " THEN 2910
2520 Y[I]=FNZ0
2530 Q[I]=FNZ0
2540 X[I]=FNZ0
2550 G[I]=FNZ0
2560 R[I]=FNZ0
2570 N$=""
2580 RETURN
2590 N=N+V-U+1
2600 GOTO (N <= 20)+2* NOT C OF 2740,2610,2640
2610 PRINT LIN1"***** NO MORE THAN 20 POINTS ALLOWED"LIN2
2620 N=N-V+U
2630 GOTO 1+C OF 2250,2360
2640 PRINT TAB20"ERRORS ARE PERCENT 2-SIGMA"LIN2,TAB80"SET";
2650 PRINT TAB22"206/238 - %ERR          207/235 - %EFR          CORREL."LIN2
2660 IF C THEN 2470
2670 FOR I=K TO N
2680 E=U+I-K
2690 Y[I]=D[E,1]+FNRE
2700 X[I]=D[E,2]
2710 Q[I]=D[E,3]
2720 G[I]=D[E,4]
2730 R[I]=D[E,5]
2740 GOTO NOT (X[I]*Y[I]*G[I]*Q[I])+2*C OF 3600,2750,2890
2750 IF (G[I]^2+Q[I]^2-2*R[I]*G[I]*Q[I]) >= 0 THEN 2780
2760 PRINT LIN1"ERROR-CORR. OF"R[I]"IS IMPOSSIBLY HIGH"LIN2
2770 GOTO 2620
2780 GOSUB 2800
2790 GOTO 2840
2800 PRINT I;N$;TAB20;Y[I];TAB32;Q[I];TAB41;X[I];TAB52;G[I];TAB63;R[I];LINC
2810 A[I]=X[I]*G[I]/100
2820 Z[I]=Y[I]*Q[I]/100
2830 RETURN
2840 IF (I+C)>20 THEN 2910
2850 I=I+C
2860 IF C THEN 2470
2870 NEXT I
2880 I=I-1
2890 PRINT LIN2
2900 N=N-C
2910 DISP "SET TO BE CORRECTED (0=OK)";
2920 INPUT I
2930 IF I=0 THEN 2980
2940 GOSUB 2490
2950 GOSUB 2800
2960 PRINT

```



```

2970 GOTO 2910
2980 FOR I=K TO N
2990 IF (X[I] >= A1)*(Y[I] >= A2)*(X[I] <= B1)*(Y[I] <= B2) THEN 3020
3000 PRINT "POINT" I "FALLS OUTSIDE PLOT LIMITS" LIN1
3010 GOTO 3080
3020 PLOT X[I],Y[I],1
3030 GOTO A-2 OF 3310,3310,3100,3100
3040 LABEL (*,1+(A=8),1+0.7*(A=8),0,P)
3050 CPLOT -0.3,-0.3
3060 LABEL (*)B$;
3070 PEN
3080 NEXT I
3090 GOTO 3390
3100 S=1-R[I]^2+1E+10*(R[I]=1)
3110 B3=-2*R[I]/(A[I]*Z[I]*S)
3120 S3=1/(S*Z[I]^2)
3130 FOR J=1 TO 1+(A=5)
3140 FOR F=-1 TO 1 STEP 2
3150 A5=-F*A[I]/(1+INT(500*A[I]/P5))
3160 FOR X=F*A[I] TO -F*A[I]+A5/2 STEP A5
3170 F1=(B3*X)^2-4*S3*(X^2/(S*A[I]^2)-1)
3180 F1=F1*(F1>0)
3190 PLOT X[I]+X,FNOF
3200 IF J=1 THEN 3230
3210 F=-F
3220 PLOT X[I]+X,FNOF
3230 NEXT X
3240 IF J=2 THEN 3070
3250 NEXT F
3260 NEXT J
3270 GOTO 3070
3280 DISP "# SIDES, ROTATION, SIZE (1-10)";
3290 INPUT B3,A9,S3
3300 GOTO 2360
3310 FOR D=0 TO 50*(A=3)
3320 B5=(S3-D/2)*P5/400
3330 IF B5<0 THEN 3070
3340 FOR J=A9 TO A9+360 STEP 360/B3
3350 PLOT X[I]+B5*COSJ,Y[I]+B5*P6*SINJ/(P*P5)
3360 NEXT J
3370 NEXT D
3380 GOTO 3070
3390 DISP ""
3400 PLOT L,P0,1
3410 IF N<2 THEN 3450
3420 DISP "YORKFIT? 1=YES";
3430 INPUT F
3440 IF F=1 THEN 3590
3450 DISP "DATA TO BE RETAINED? 1=YES";
3460 INPUT F

```

```
3470 IF F#1 THEN 2250
3480 K=N+1
3490 GOTO 2290
3500 DISP "LETTER SIZE (.5-2), ANGLE";
3510 INPUT F,F1
3520 LABEL (*,2*F3/3,1.7,F1,P)
3530 DISP "POSITION PEN- PRESS STOP WHEN OK"
3540 LETTER
3550 DISP "PHRASE";
3560 INPUT A$
3570 LABEL (*)A$
3580 GOTO 2250
3590 LINK F[2]
3600 PRINT LIN1"***IMPROPER INPUT- TRY AGAIN***"LIN2
3610 GOTO 2620
3620 END
```

Second Segment:

```

10 REM PART 2 (YORKFIT & INTERCEPTS) OF CONCORDIA PLOTTER FOR 8K 9830/9862
11 REM K.R. LUDWIG, 1982
20 FIND F[1]
30 PRINT TAB18"***** YORKFIT *****"LIN2
40 PRINT TAB21"ERRORS ARE PERCENT 2-SIGMA"TAB90
50 PRINT "SET#      206/238 -- %ERR"TAB28"207/235 -- %ERR"TAB51"7/6-%ERR";
60 PRINT TAB64"CORREL."LIN2
70 FORMAT F3.0,F12.5,F8.2,F11.4,F8.2,6X,F9.3,5X,F8.4
110 I3=0
120 Q=0.1
130 S3=Q
140 I1=0
150 Q=S3
160 I1=1+I1
170 IF I1>10 THEN 2360
174 IF I1>1 AND I3=0 THEN 290
180 FOR I=1 TO N
190 IF I3=0 THEN 240
200 G[I]=1
210 Q[I]=1/Q^2
220 R[I]=0
230 GOTO 275
240 WRITE (15,70)I,Y[I],Q[I],X[I],G[I],SQR(Q[I]^2+G[I]^2-2*R[I]*Q[I]*G[I]),R[I]
260 Q[I]=4/Z[I]^2
270 G[I]=4/A[I]^2
275 A[I]=SQR(G[I]*Q[I])
280 NEXT I
290 S1=B5=D=E=X1=Y1=0
300 FOR I=1 TO N
320 Z[I]=G[I]*Q[I]/(Q^2*Q[I]+G[I]-2*Q*R[I]*A[I])
330 S1=S1+Z[I]
340 X1=X1+Z[I]*X[I]
350 Y1=Y1+Z[I]*Y[I]
360 NEXT I
370 X1=X1/S1
380 Y1=Y1/S1
390 FOR I=1 TO N
400 U=X[I]-X1
410 V=Y[I]-Y1
420 B5=B5+(U^2/Q[I]-V^2/G[I])*Z[I]^2
430 D=D+(U*V/G[I]-R[I]*U^2/A[I])*Z[I]^2
440 E=E+V*(U/Q[I]-R[I]*V/A[I])*Z[I]^2
450 NEXT I
460 S3=B5^2+4*D*E

```

```

470 IF S3<0 THEN 2360
480 S3=(SQRS3-B5)/(2*D)
500 IF ABS(S3-Q)/S3>1E-04 THEN 150
510 Q=S3
520 A7=Y1-Q*X1
530 D=E=S=0
534 REM ERROR ALGORITHM OF TITTERINGTON & HALLIDAY, CHEM GEOL V 26, P 183
540 FOR I=1 TO N
545 Y=Y[I]-Q*X[I]-A7
550 S=S+Z[I]*Y^2
560 X=X[I]-Z[I]*Y*(R[I]*A[I]-Q*Q[I])/(Q[I]*G[I])
570 D=D+X*Z[I]
580 E=E+X^2*Z[I]
590 NEXT I
591 V=N-2
592 K=(V<2)*12.7+(V>1)*1.96*V/SQR(V^2-2.43*V+1.696)
600 M=(V#0)*S/(V+NOT V)
610 B3=SQR(S1/(E*S1-D^2))
620 F1=B3*SQR(E/S1)
630 S1=F1*SQRM
640 B5=B3*SQRM
650 IF I3 THEN 980
660 IF S#0 AND ((S<13)+(N>9)*(S<22)+(N>14)*(S<30)) THEN 690
670 A=NOT S
680 GOTO 890
690 D=A=1
710 F=(INT(V/2)#V/2)
720 IF NOT F THEN 840
730 FOR I=2 TO 2000 STEP 2
740 D=D*S/(V+I)
750 A=A+D
760 IF D<1E-06 THEN 780
770 NEXT I
780 D=SQRPI
790 FOR I=0.5 TO V/2
800 D=D*I
810 NEXT I
820 A=1-EXP(-S/2)*(S/2)^(V/2)*A/D
830 GOTO 890
840 FOR I=1 TO V/2-1
850 D=D*S/(2*I)
860 A=A+D
870 NEXT I
880 A=A*EXP(-S/2)
890 I0=1+(A<0.01)+(A<0.3)
900 WRITE (15,1060)
910 PRINT "***MODEL 1 SOLUTION -- ASSUMES ALL SCATTER DUE TO ANALYTICAL ERROR**"
920 PRINT
930 I5=P2=1.96*B3
940 I4=P1=1.96*F1
950 Q2=Q
960 A6=A7

```

```

962 Y9=Y1
964 X9=X1
970 GOTO 1030
980 PRINT LIN1"*****MODEL 2 SOLUTION -- EQUALLY WEIGHTED POINTS, ASSUMES"
990 PRINT "*****NORMAL DISTRIBUTION OF RESIDUALS*****"
1000 P2=K*B5
1010 P1=K*S1
1020 PRINT TAB90
1030 FORMAT 33X,"SLOPE",19X,"INTERCEPT",/,33X,5"-",19X,9"-",/
1040 WRITE (15,1030)
1050 PRINT "BEST-FIT:"TAB31,Q;TAB56;A7
1060 FORMAT 72"- "
1070 WRITE (15,1060)
1080 IF I3 THEN 1100
1090 PRINT "ERROR (1-SIGMA, A PRIORI) "TAB31,B3;TAB56;F1,LIN1
1100 PRINT "ERROR (1-SIGMA, FROM SCATTER) "TAB31,B5;TAB56;S1,LIN1
1110 PRINT "ERROR (95% CONFIDENCE LIMITS) "TAB31,P2;TAB56;P1,LIN1
1120 PRINT "COORDINATES OF CENTROID: X="X1;"Y="Y1,LIN1
1130 IF I3 THEN 1190
1140 FIXED 2
1150 PRINT "ANALYTICAL ERRORS ALONE WILL CAUSE AT LEAST THE OBSERVED AMOUNT OF"
1160 PRINT "SCATTER"100*A"% OF THE TIME, AS INDICATED BY THE M.S.W.D. OF"M
1185 STANDARD
1190 WRITE (15,1060)
1200 I3=1+I3
1210 IF I3#2 AND I0>1 THEN 130
1220 IF I0=2 THEN 1250
1230 I0=(I0=1)+2*(I0=3)
1240 GOTO 1270
1250 DISP "MODEL 1 OR MODEL 2 SOLUTION";
1260 INPUT I0
1270 Q=Q2*(I0=1)+Q*(I0=2)
1280 A7=A6*(I0=1)+A7*(I0=2)
1290 P2=P2*(I0=2)+I5*(I0=1)
1300 Y1=Y1*(I0=2)+Y9*(I0=1)
1302 X1=X1*(I0=2)+X9*(I0=1)
1304 P1=P1*(I0=2)+I4*(I0=1)
1310 F=((A1*Q+A7)<A2)+((B2-A7)/Q>B1)/2
1320 PLOT ((A2-A7)/Q)*INTF+A1* NOT INTF,A2*INTF+(A1*Q+A7)* NOT INTF
1330 PLOT B1*(INTF<F)+((B2-A7)/Q)*(INTF=F),(Q*B1+A7)*(INTF<F)+B2*(INTF=F),-1
1340 FOR J=0 TO 1
1350 T1=-500+5500*J
1360 X=(A7+FNCT1*FNAT1-FNBT1)/(FNCT1-Q)
1370 IF X<-1 THEN 2260
1380 T=LOG(1+X)/L5
1390 IF ABS(T-T1)<0.01 THEN 1420
1400 T1=T
1410 GOTO 1360
1420 G[J+1]=T
1430 NEXT J
1440 RAD

```

```

1500 FOR J=3 TO 6
1510 T1=7000*(J>4)-1000
1520 A6=1-2*(J/2#INT(J/2))
1530 I3=9E+99
1540 GOSUB 1580
1550 G[J]=T
1560 NEXT J
1570 GOTO 1720
1580 S1=FNCT1
1590 B5=FNBT1-S1*FNAT1-A7
1600 D=2*(B5*(S1-Q)+X1*P2^2)
1610 E=(S1-Q)^2-P2^2
1620 I1=D^2-4*E*(B5^2-P1^2)
1630 IF I1<0 THEN 2290
1640 X=(A6*SQR I1-D)/(2*E)
1650 IF X<-1 THEN 2290
1660 T=LOG(1+X)/L5
1670 GOTO (ABS(T-T1)>I3)+2*(ABS(T-T1)<0.01) OF 2290,1710,2290
1680 I3=ABS(T-T1)
1690 T1=T
1700 GOTO 1580
1710 RETURN
1720 G[8]=ABS((G[4]-G[3])/2)*((G[3]*G[4])#0)
1730 G[9]=ABS((G[6]-G[5])/2)*((G[6]*G[5])#0)
1740 FORMAT 9"<"," MODEL",F2.0," CONCORDIA-INTERCEPT SOLUTION ",9">","/,/
1750 WRITE (15,1740)I0
1760 FORMAT "LOWER INTERCEPT: ",F7.1," +/-",F5.0," M.Y. (>",F5.0," <",F5.0,"
1770 FORMAT "UPPER INTERCEPT: ",F7.1," +/-",F5.0," M.Y. (>",F5.0," <",F5.0,"
1780 WRITE (15,1760)G[1],G[8],G[3],G[4]
1790 WRITE (15,1770)G[2],G[9],G[5],G[6]
1800 FORMAT "(UNCERTAINTIES ARE 95% CONFIDENCE-LIMITS)",/,72"-",/,/
1810 WRITE (15,1800)
1820 DEG
1830 PLOT B1,A2,1
1840 LABEL (*,F3/1.3,1.7,0,P)
1850 CPLOT -4,3+Q1*3
1860 CPLOT -4,0
1870 CPLOT -8+3* NOT G[8]+3* NOT G[9],0
1880 CPLOT -7.5,0
1890 Q1=1+Q1
1900 LABEL (*)"INTERCEPTS AT";
1910 CPLOT -7.5+2.5* NOT G[8]+2.5* NOT G[9],-1.1
1920 CPLOT -6,0
1921 CPLOT -5,0
1930 FOR I=1 TO 2
1940 FIXED 0
1950 LABEL (*)G[I];
1960 IF NOT G[I+7] THEN 2040

```

```

1970 CPLOT FNII-3*(FNII>3)-4.3,0.12
1980 LABEL (*)"+";
1990 CPLOT -1,-0.48
2000 LABEL (*)"-";
2010 CPLOT -0.3,0.36
2020 FIXED 0
2030 LABEL (*)G[I+7];
2040 CPLOT FNI(I+7)-3*(FNI(I+7)>3)-3.8+2*(FNII<4)* NOT G[I+7],0
2050 IF I=2 THEN 2080
2060 LABEL (*)"AND";
2070 NEXT I
2080 LABEL (*)"M.Y.";
2090 PLOT L,P0,1
2100 DISP ""
2110 DISP "ENTER 1 FOR ERROR-ENVELOPE PLOT";
2120 INPUT E
2130 IF E#1 THEN 2400
2140 FOR I=1 TO 2
2150 FOR X=A1 TO B1 STEP Z6/110
2160 F=(P1^2+P2^2*X*(X-2*X1))
2170 IF F<0 OR X<A1 THEN 2210
2180 F=A7+X*Q+(3-2*I)*SQRF
2190 GOTO (F<A2)+2*(F>B2) OF 2210,2220
2200 PLOT X,F
2210 NEXT X
2220 PEN
2230 NEXT I
2240 GOTO 2400
2260 PRINT "***** CHORD DOES NOT HAVE TWO CONCORDIA INTERCEPTS *****"LIN1
2270 G[1+J]=0
2280 GOTO 1430
2290 T=0
2310 PRINT "ARM"J-2"OF ERROR-ENVELOPE HAS NO CONCORDIA INTERCEPT"LIN1
2330 G[J]=0
2340 GOTO 1710
2360 PRINT LIN1"DATA CANNOT BE FIT TO A STRAIGHT LINE"LIN2
2370 GOTO 1720
2400 LINK F[1],10,40
2410 DEF FNA(T)=EXP(L5*T)-1
2420 DEF FNB(T)=EXP(L8*T)-1
2430 DEF FNC(T)=L8*EXP((L8-L5)*T)/L5
2450 DEF FNI(X)=1+INTLG(TABS(INTG[X])+ NOT (INTG[X]))
2560 END

```